



---

# *ABEL Design Manual*

---

Version 8.0

35 pages about ispDesignExpert removed

<b>Chapter 4 ABEL-HDL Design Considerations</b> .....	<b>37</b>
Overview of ABEL-HDL Design Considerations .....	38
Hierarchy in ABEL-HDL .....	38
Instantiating a Lower-level Module in an ABEL-HDL Source .....	39
Identifying I/O Ports in the Lower-level Module .....	39
Declaring Lower-level Modules in the Top-level Source .....	40
Instantiating Lower-level Modules in Top-level Source .....	40
Hierarchy and Retargeting and Fitting .....	41
Redundant Nodes .....	41
Merging Feedbacks .....	41
Post-linked Optimization .....	41
Hierarchical Design Considerations .....	42
Prevent Node Collapsing .....	42
Node Collapsing .....	42
Selective Collapsing .....	42
Pin-to-pin Language Features .....	43
Device-independence vs. Architecture-independence .....	43
Signal Attributes .....	43
Signal Dot Extensions .....	43
Pin-to-pin vs. Detailed Descriptions for Registered Designs .....	44
Using := for Pin-to-pin Descriptions .....	44
Resolving Ambiguities .....	44
Detailed Circuit Descriptions .....	45
Detailed Descriptions: Designing for Macrocells .....	45
Examples of Pin-to-pin and Detailed Descriptions .....	47
Pin-to-pin Module Description .....	47
Detailed Module Description .....	47
Detailed Module with Inverted Outputs .....	48
When to Use Detailed Descriptions .....	50
Using := for Alternative Flip-flop Types .....	50
Using Active-low Declarations .....	51
Polarity Control .....	53
Polarity Control with Istype .....	53
Using Istype 'neg', 'pos', and 'dc' to Control Equation and Device Polarity .....	53
Using 'invert' and 'buffer' to Control Programmable Inversion .....	54
Flip-flop Equations .....	54
Feedback Considerations — Dot Extensions .....	55
Dot Extensions and Architecture-Independence .....	56
Dot Extensions and Detail Design Descriptions .....	58
Using Don't Care Optimization .....	60
Exclusive OR Equations .....	62
Optimizing XOR Devices .....	62
Using XOR Operators in Equations .....	62
Using Implied XORs in Equations .....	62
Using XORs for Flip-flop Emulation .....	63
JK Flip-Flop Emulation .....	63

---

State Machines	65
Use Identifiers Rather Than Numbers for States	65
Powerup Register States	67
Unsatisfied Transition Conditions	67
D-Type Flip-Flops	67
Other Flip-flops	68
Precautions for Using Don't Care Optimization	68
Number Adjacent States for One-bit Change	72
Use State Register Outputs to Identify States	72
State Register Bit Values	73
Using Symbolic State Descriptions	74
Symbolic Reset Statements	74
Symbolic Test Vectors	75
Using Complement Arrays	75
ABEL-HDL and Truth Tables	77
Basic Syntax - Simple Examples	78
Influence of Signal polarity	79
Using .X. in Truth tables conditions	80
Using .X. on the right side	81
Special case: Empty ON-set	82
Registered Logic in Truth tables	82

## Chapter 4 ***ABEL-HDL Design Considerations***

---

This chapter covers the following topics:

- Overview of ABEL-HDL Design Considerations
- Hierarchy in ABEL-HDL
- Hierarchical Design Considerations
- Node Collapsing
- Pin-to-pin Language Features
- Pin-to-pin vs. Detailed Descriptions for Registered Designs
- Using Active-low Declarations
- Polarity Control
- Flip-flop Equations
- Feedback Considerations — Dot Extensions
- Using Don't Care Optimization
- Exclusive OR Equation
- State Machines
- Using Complement Arrays
- ABEL-HDL and Truth Tables

## Overview of ABEL-HDL Design Considerations

This chapter discusses issues you need to consider when you create a design with ABEL-HDL. The topics covered are listed below:

- Hierarchy in ABEL-HDL
- Hierarchical Design Considerations
- Node Collapsing
- Pin-to-Pin Architecture-independent Language Features
- Pin-to-Pin Vs. Detailed Descriptions for Registered Designs
- Using Active-low Declarations
- Polarity Control
- Istyles and Attributes
- Flip-flop Equations
- Feedback Considerations — Using Dot Extensions
- @DCSET Considerations and Precautions
- Exclusive OR Equations
- State Machines
- Using Complement Arrays
- ABEL-HDL and Truth Tables

## Hierarchy in ABEL-HDL

You use hierarchy declarations in an upper-level ABEL-HDL source to refer to (instantiate) an ABEL-HDL module.

*To instantiate an ABEL-HDL module:*

In the lower-level module: (optional)

1. Identify lower-level I/O Ports (signals) with an INTERFACE statement.

In the top-level source:

2. Declare the lower-level module with an INTERFACE declaration.
3. Instantiate the lower-level module with FUNCTIONAL\_BLOCK declarations.

**NOTE**

Hierarchy declarations are not required when instantiating an ABEL-HDL module in a schematic. For instructions on instantiating lower-level modules in schematics, refer to your schematic reference.

## Instantiating a Lower-level Module in an ABEL-HDL Source

### Identifying I/O Ports in the Lower-level Module

The way to identify an ABEL-HDL module's input and output ports is to place an INTERFACE statement immediately following the MODULE statement. The INTERFACE statement defines the ports in the lower-level module that are used by the top-level source.

You must declare all input pins in the ABEL-HDL module as ports, and you can specify default values of 0, 1, or Don't-care.

You do not have to declare all output pins as ports. Any undeclared outputs become No Connects or redundant nodes. Redundant nodes can later be removed from the designs during post-link optimization.

The following source fragment is an example of a lower-level INTERFACE statement.

```
module lower
interface (a=0, [d3..d0]=7 -> [z0..z7]) ;
title 'example of lower-level interface statement ' ...
```

This statement identifies input a, d3, d2, d1 and d0 with default values, and outputs z0 through z7. For more information, see “Interface (lower-level)” in the [ABEL-HDL Reference Manual](#).

### Specifying Signal Attributes

Attributes specified for pins in a lower-level module are propagated to the higher-level source. For example, a lower-level pin with an ‘invert’ attribute affects the higher-level signal wired to that pin (it affects the pin's preset, reset, preload, and power-up value).

### Output Enables (OE)

Connecting a lower-level tristate output to a higher-level pin results in the output enable being specified for the higher-level pin. If another OE is specified for the higher-level pin, it is flagged as an error. Since most tristate outputs are used as bidirectionals, it might be important to keep the lower-level OE.

## Buried Nodes

Buried nodes in lower-level sources are handled as follows:

Dangling Nodes	Lower-level nodes that do not fanout are propagated to the higher-level module and become dangling nodes. Optimization may remove dangling nodes.
Combinational nodes	Combinational nodes in a lower-level module become collapsible nodes in the higher-level module.
Registered nodes	Registered nodes are preserved with hierarchical names assigned to them.

## Declaring Lower-level Modules in the Top-level Source

To declare a lower-level module, you match the lower-level module's INTERFACE statement with an INTERFACE declaration. For example, to declare the lower-level module given above, you would add the following declaration to your upper-level source declarations:

```
lower interface (a, [d3..d0] -> [z0..z7]) ;
```

You could specify different default values if you want to override the values given in the instantiated module, otherwise the instantiated module must exactly match the lower-level interface statement. See "Interface (top-level)" in the [ABEL-HDL Reference Manual](#) for more information.

## Instantiating Lower-level Modules in Top-level Source

Use a FUNCTIONAL\_BLOCK declaration in an top-level ABEL-HDL source to instantiate a declared lower-level module and make the ports of the lower-level module accessible in the upper-level source. You must declare sources with an INTERFACE declaration before you instantiate them.

To instantiate the module declared above, add an interface declaration and signal declarations to your top-level declarations, and add port connection equations to your top-level equations, as shown in the source fragment below:

```
DECLARATIONS
  low1 FUNCTIONAL_BLOCK lower ;
  zed0..zed7 pin ;                "upper-level inputs
  atop pin istype 'reg,buffer';   "upper-level output
  d3..d0 pin istype 'reg,buffer'; "upper-level ouputs
EQUATIONS
  atop = low1.a;                  "wire this source's outputs
  [d3..d0] = low1.[d3..d0] ;      "to lower-level inputs
  low1.[z0..z7] = [zed0..zed7];   "wire this source's inputs
                                   "to lower-level outputs
```

See "Functional\_block" in the [ABEL-HDL Reference Manual](#) for more information.

## Hierarchy and Retargeting and Fitting

### Redundant Nodes

When you link multiple sources, some unreferenced nodes may be generated. These nodes usually originate from lower-level outputs that are not being used in the top-level source. For example, when you use a 4-bit counter as a 3-bit counter. The most significant bit of the counter is unused and can be removed from the design to save device resources. This step also removes trivial connections. In the following example, if `out1` is a pin and `t1` is a node:

```
out1 = t1;  
t1 = a86;
```

would be mapped to

```
out1 = a86;
```

### Merging Feedbacks

Linking multiple modules can produce signals with one or more feedback types, such as `.FB` and `.Q`. You can tell the optimizer to combine these feedbacks to help the fitting process.

### Post-linked Optimization

If your design has a constant tied to an input, you can re-optimize the design. Re-optimizing may further reduce the product terms count.

For example, if you have the equation

```
out = i0 & i1 || !i0 & i2;
```

and `i0` is tied to 1, the resulting equation would be simplified to

```
out = i1;
```



# Hierarchical Design Considerations

The following considerations apply to hierarchical design.

## Prevent Node Collapsing

Use the signal attribute 'keep' to indicate that the combinational node should not be collapsed (removed). For example, the following ABEL-HDL source uses the 'keep' signal attribute:

```
MODULE sub1
TITLE 'sub-module 1'
a,b,c pin;
d pin ;
e node istype 'keep';
Equations
e = a $ b;
d = c & e;
END
```

## Node Collapsing

All combinational nodes are collapsible by default. Nodes that are to be collapsed (or nodes that are to be preserved) are flagged through the use of signal attributes in the language. The signal attributes are:

Istype 'keep'	Do not collapse this node.
'collapse'	Collapse this node.

Collapsing provides multi-level optimization for combinational logic. Designs with arithmetic and comparator circuits generally generate a large number of product terms that will not fit to any programmable logic device. Node collapsing allows you to describe equations in terms of multi-level combinational nodes, then collapse the nodes into the output until it reaches the product term you specify. The result is an equation that is optimized to fit the device constraints.

## Selective Collapsing

In some instances you may want to prevent the collapsing of certain nodes. For example, some nodes may help in the simulation process. You can specify nodes you do not want collapsed as Istype 'keep' and the optimizer will not collapse them.

## Pin-to-pin Language Features

ABEL-HDL is a device-independent language. You do not have to declare a device or assign pin numbers to your signals until you are ready to implement the design into a device. However, when you do not specify a device or pin numbers, you need to specify pin-to-pin attributes for declared signals.

Because the language is device-independent, the ABEL-HDL compiler does not have predetermined device attributes to imply signal attributes. If you do not specify signal attributes or other information (such as the dot extensions, which are described later), your design might not operate consistently if you later transfer it to a different target device.

### Device-independence vs. Architecture-independence

The requirement for signal attributes does not mean that a complex design must always be specified with a particular device in mind. You may still have to understand the differences between GAL devices and ispLSI devices, but you do not have to specify a particular device when describing your design.

Attributes and dot extensions help you refine your design to work consistently when moving from one class of device architecture to another; for example from devices having inverted outputs to those with a particular kind of reset/preset circuitry. However, the more you refine your design, using these language features, the more restrictive your design becomes in terms of the number of device architectures for which it is appropriate.

### Signal Attributes

Signal attributes remove ambiguities that occur when no specific device architecture is declared. If your design does not use device-related attributes (either implied by a DEVICE statement or expressed in an ISTYPE statement), it may not operate the same way when targeted to different device architectures. See “Pin Declaration,” “Node Declaration” and “Istype” in the [ABEL-HDL Reference Manual](#) for more information.

### Signal Dot Extensions

Signal dot extensions, like attributes, enable you to more precisely describe the behavior of a circuit that may be targeted to different architectures. Dot extensions remove the ambiguities in equations.

Refer to [“Feedback Considerations — Dot Extensions” on page 55](#) and “Language Structure” in the [ABEL-HDL Reference Manual](#) for more information.

## Pin-to-pin vs. Detailed Descriptions for Registered Designs

You can use ABEL-HDL assignment operators when you write high-level equations. The = operator specifies a combinational assignment, where the design is written with only the circuit's inputs and outputs in mind. The := assignment operator specifies a registered assignment, where you must consider the internal circuit elements (such as output inverters, presets and resets) related to the memory elements (typically flip-flops). The semantics of these two assignment operators are discussed below.

### Using := for Pin-to-pin Descriptions

The := implies that a memory element is associated with the output defined by the equation. For example, the equation;

```
Q1 := !Q1 # Preset;
```

implies that Q1 will hold its current value until the memory element associated with that signal is clocked (or unlatched, depending on the register type). This equation is a pin-to-pin description of the output signal Q1. The equation describes the signal's behavior in terms of desired output pin values for various input conditions. Pin-to-pin descriptions are useful when describing a circuit that is completely architecture-independent.

Language elements that are useful for pin-to-pin descriptions are the “:=” assignment operator, and the .CLK, .OE, .FB, .CLR, .ACLR, .SET, .ASET and .COM dot extensions described in the [ABEL-HDL Reference Manual](#). These dot extensions help resolve circuit ambiguities when describing architecture-independent circuits.

### Resolving Ambiguities

In the equation above (Q1 := !Q1 # Preset;), there is an ambiguous feedback condition. The signal Q1 appears on the right side of the equation, but there is no indication of whether that feedback signal should originate at the register, come directly from the combinational logic that forms the input to the register, or come from the I/O pin associated with Q1. There is also no indication of what type of register should be used (although register synthesis algorithms could, theoretically, map this equation into virtually any register type). The equation could be more completely specified in the following manner:

```
Q1.CLK = Clock;          "Register clocked from input
Q1 := !Q1.FB # Preset;  "Reg. feedback normalized to pin value
```

This set of equations describes the circuit completely and specifies enough information that the circuit will operate identically in virtually any device in which you can fit it. The feedback path is specified to be from the register itself, and the .CLK equation specifies that the memory element is clocked, rather than latched.

## Detailed Circuit Descriptions

In contrast to a pin-to-pin description, the same circuit can be specified in a detailed form of design description in the following manner:

```
Q1.CLK = Clock;           "Register clocked from input
Q1.D   = !Q1.Q # Preset;  "D-type f/f used for register
```

In this form of the design, specifying the D input to a D-type flip-flop and specifying feedback directly from the register restricts the device architectures in which the design can be implemented. Furthermore, the equations describe only the inputs to, and feedback from, the flip-flop and do not provide any information regarding the configuration of the actual output pin. This means the design will operate quite differently when implemented in a device with inverted outputs versus a device with non-inverting outputs.

To maintain the correct pin behavior, using detailed equations, one additional language element is required: a 'buffer' attribute (or its complement, an 'invert' attribute). The 'buffer' attribute ensures that the final implementation in a device has no inversion between the specified D-type flip-flop and the output pin associated with Q1. For example, add the following to the declarations section:

```
Q1 pin istype 'buffer';
```

### Detailed Descriptions: Designing for Macrocells

One way to understand the difference between pin-to-pin and detailed description methods is to think of detailed descriptions as macrocell specifications. A macrocell is a block of circuitry normally (but not always) associated with a device's I/O pin. Figure 4-1 illustrates a typical macrocell associated with signal Q1.

*Detailed Macrocell*




Figure 4-1. Detailed Macrocell

Detailed descriptions are written for the various input ports of the macrocell (shown in the figure above with dot extension labels). Note that the macrocell features a configurable inversion between the Q output of the flip-flop and the output pin labeled Q1. If you use this inverter (or select a device that features a fixed inversion), the behavior you observe on the Q1 output pin will be inverted from the logic applied to (or observed on) the various macrocell ports, including the feedback port Q1.q.

Pin-to-pin descriptions, on the other hand, allow you to describe your circuit in terms of the expected behavior on an actual output pin, regardless of the architecture of the underlying macrocell. Figure 4-2 illustrates the pin-to-pin concept:

*Pin-to-pin Macrocell*




Figure 4-2. Pin-to-pin Macrocell

When pin-to-pin descriptions are written in ABEL-HDL, the “generic macrocell” shown above is synthesized from whatever type of macrocell actually exists in the target device.

## Examples of Pin-to-pin and Detailed Descriptions

Two equivalent module descriptions, one pin-to-pin and one detailed, are shown below for comparison:

### Pin-to-pin Module Description

```

module Q1_1
    Q1                pin        istype 'reg';
    Clock,Preset      pin;

    equations
        Q1.clk = Clock;
        Q1      := !Q1.fb # Preset;

    test_vectors ([Clock,Preset] -> Q1)
        [ .c. , 1 ] -> 1;
        [ .c. , 0 ] -> 0;
        [ .c. , 0 ] -> 1;
        [ .c. , 0 ] -> 0;
        [ .c. , 1 ] -> 1;
        [ .c. , 1 ] -> 1;
end

```

### Detailed Module Description

```

module Q1_2
    Q1                pin        istype 'reg_D,buffer';
    Clock,Preset      pin;

    equations
        Q1.CLK = Clock;
        Q1.D   = !Q1.Q # Preset;

    test_vectors ([Clock,Preset] -> Q1)
        [ .c. , 1 ] -> 1;
        [ .c. , 0 ] -> 0;
        [ .c. , 0 ] -> 1;
        [ .c. , 0 ] -> 0;
        [ .c. , 1 ] -> 1;
        [ .c. , 1 ] -> 1;
end

```

The first description can be targeted into virtually any device (if register synthesis and device fitting features are available), while the second description can be targeted only to devices featuring D-type flip-flops and non-inverting outputs.

To implement the second (detailed) module in a device with inverting outputs, the source file would need to be modified as shown in the following section.

## Detailed Module with Inverted Outputs

```

module Q1_3
    Q1                pin        istype 'reg_D,invert';
    Clock,Preset     pin;

equations
    Q1.CLK = Clock;
    !Q1.D   = Q1.Q # Preset;

test_vectors ([Clock,Preset] -> Q1)
    [ .c. , 1 ] -> 1;
    [ .c. , 0 ] -> 0;
    [ .c. , 0 ] -> 1;
    [ .c. , 0 ] -> 0;
    [ .c. , 1 ] -> 1;
    [ .c. , 1 ] -> 1;
end

```

In this version of the module, the existence of an inverter between the output of the D-type flip-flop and the output pin (specified with the 'invert' attribute) has necessitated a change in the equation for Q1.D.

As this example shows, device-independence and pin-to-pin description methods are preferable, since you can describe a circuit completely for any implementation. Using pin-to-pin descriptions and generalized dot extensions (such as `.FB`, `.CLK` and `.OE`) as much as possible allows you to implement your ABEL-HDL module into any one of a particular class of devices. (For example, any device that features enough flip-flops and appropriately configured I/O resources.) However, the need for particular types of device features (such as register preset or reset) might limit your ability to describe your design in a completely architecture-independent way.

If, for example, a built-in register preset feature is used in a simple design, the target architectures are limited. Consider this version of the design:

```

module Q1_51
    Q1                pin        istype 'reg,buffer';
    Clock,Preset     pin;

equations
    Q1.CLK = Clock;
    Q1.AP  = Preset;
    Q1     := !Q1.fb ;

test_vectors ([Clock,Preset] -> Q1)
    [ .c. , 1 ] -> 1;
    [ .c. , 0 ] -> 0;
    [ .c. , 0 ] -> 1;
    [ .c. , 0 ] -> 0;
    [ .c. , 1 ] -> 1;
    [ .c. , 1 ] -> 1;

end

```

The equation for Q1 still uses the := assignment operator and .FB for a pin-to-pin description of Q1's behavior, but the use of .AP to describe the reset function requires consideration of different device architectures. The .AP extension, like the .D and .Q extensions, is associated with a flip-flop input, not with a device output pin. If the target device has inverted outputs, the design will not reset properly, so this ambiguous reset behavior is removed by using the 'buffer' attribute, which reduces the range of target devices to those with non-inverted outputs.

Using .ASET instead of .AP can solve this problem if the fitter being used supports the .ASET dot extension.

Versions 5 and 7 of the design above and below are unambiguous, but each is restricted to certain device classes:

```

module Q1_71
    Q1                pin        istype 'reg,invert';
    Clock,Preset     pin;

equations
    Q1.CLK = Clock;
    Q1.AR  = Preset;
    Q1     := !Q1.fb ;

test_vectors ([Clock,Preset] -> Q1)
    [ .c. , 1 ] -> 1;
    [ .c. , 0 ] -> 0;
    [ .c. , 0 ] -> 1;
    [ .c. , 0 ] -> 0;
    [ .c. , 1 ] -> 1;
    [ .c. , 1 ] -> 1;

end

```



## When to Use Detailed Descriptions

Although the pin-to-pin description is preferable, there will frequently be situations when you must use a more detailed description. If you are unsure about which method to use for various parts of your design, examine the design's requirements. If your design requires specific features of a device (such as register preset or unusual flip-flop configurations), detailed descriptions are probably necessary. If your design is a simple combinational function, or if it matches the "generic" macrocell in its requirements, you can probably use simple pin-to-pin descriptions.

## Using := for Alternative Flip-flop Types

In ABEL-HDL you can specify a variety of flip-flop types using attributes such as `istype 'reg_D'` and `'reg_JK'`. However, these attributes do not enforce the use of a specific type of flip-flop when a device is selected, and they do not affect the meaning of the `:=` assignment operator.

You can think of the `:=` assignment operator as a memory operator. The type of register that most closely matches the `:=` assignment operator's behavior is the D-type flip-flop.

The primary use for attributes such as `istype 'reg_D'`, `'reg_JK'` and `'reg_SR'` is to control the generation of logic. Specifying one of the `'reg_'` attributes (for example, `istype 'reg_D'`) instructs the AHDL compiler to generate equations using the `.D` extension regardless of whether the design was written using `.D`, `:=` or some other method (for example, state diagrams).



### **NOTE**

You also need to specify `istype 'invert'` or `'buffer'` when you use detailed syntax.

Using `:=` for flip-flop types other than D-type is only possible if register synthesis features are available to convert the generated equations into equations appropriate for the alternative flip-flop type specified. Since the use of register synthesis to convert D-type flip-flop stimulus into JK or SR-type stimulus usually results in inefficient circuitry, the use of `:=` for these flip-flop types is discouraged. Instead, you should use the `.J` and `.K` extensions (for JK-type flip-flops) or the `.S` and `.R` extensions (for SR-type flip-flops) and use a detailed description method (including `'invert'` or `'buffer'` attributes) to describe designs for these register types.

There is no provision in the language for directly writing pin-to-pin equations for registers other than D-type. State diagrams, however, may be used to describe pin-to-pin behavior for any register type.

## Using Active-low Declarations

In ABEL-HDL you can write pin-to-pin design descriptions using implied active-low signals. Active-low signals are declared with a '!' operator, as shown below:

```
!Q1 pin  istype 'reg';
```

If a signal is declared active-low, it is automatically complemented when you use it in the subsequent design description. This complementing is performed for any use of the signal itself, including as an input, as an output, and in test vectors. Complementing is also performed if you use the `.fb` dot extension on an active-low signal.

The following three designs, for example, operate identically:

### Design 1 — Implied Pin-to-Pin Active-low

```
module act_low2
    !q0,!q1  pin istype 'reg';
    clock    pin;
    reset    pin;

    equations
        [q1,q0].clk = clock;
        [q1,q0] := ([q1,q0].FB + 1) & !reset;

    test_vectors ([clock,reset] -> [ q1, q0])
        [ .c. , 1 ] -> [ 0 , 0 ];
        [ .c. , 0 ] -> [ 0 , 1 ];
        [ .c. , 0 ] -> [ 1 , 0 ];
        [ .c. , 0 ] -> [ 1 , 1 ];
        [ .c. , 0 ] -> [ 0 , 0 ];
        [ .c. , 0 ] -> [ 0 , 1 ];
        [ .c. , 1 ] -> [ 0 , 0 ];

end
```

**Design 2 — Explicit Pin-to-Pin Active-low**

```

module act_low1
    q0,q1    pin istype 'reg';
    clock    pin;
    reset    pin;

    equations
        [q1,q0].clk = clock;
        ![q1,q0] := (![q1,q0].FB + 1) & !reset;

    test_vectors ([clock,reset] -> [!q1,!q0])
        [ .c. , 1 ] -> [ 0 , 0 ];
        [ .c. , 0 ] -> [ 0 , 1 ];
        [ .c. , 0 ] -> [ 1 , 0 ];
        [ .c. , 0 ] -> [ 1 , 1 ];
        [ .c. , 0 ] -> [ 0 , 0 ];
        [ .c. , 0 ] -> [ 0 , 1 ];
        [ .c. , 1 ] -> [ 0 , 0 ];

end

```

**Design 3 — Explicit Detailed Active-low**

```

module act_low3
    q0,q1    pin istype 'reg_d,buffer';
    clock    pin;
    reset    pin;

    equations
        [q1,q0].clk = clock;
        ![q1,q0].D := (![q1,q0].Q + 1) & !reset;

    test_vectors ([clock,reset] -> [!q1,!q0])
        [ .c. , 1 ] -> [ 0 , 0 ];
        [ .c. , 0 ] -> [ 0 , 1 ];
        [ .c. , 0 ] -> [ 1 , 0 ];
        [ .c. , 0 ] -> [ 1 , 1 ];
        [ .c. , 0 ] -> [ 0 , 0 ];
        [ .c. , 0 ] -> [ 0 , 1 ];
        [ .c. , 1 ] -> [ 0 , 0 ];

end

```

Both of these designs describe an up counter with active-low outputs. The first example inverts the signals explicitly (in the equations and in the test vector header), while the second example uses an active-low declaration to accomplish the same thing.

## Polarity Control

Automatic polarity control is a powerful feature in ABEL-HDL where a logic function is converted for both non-inverting and inverting devices.

A single logic function may be expressed with many different equations. For example, all three equations below for F1 are equivalent.

- (1)  $F1 = (A \& B);$
- (2)  $!F1 = !(A \& B);$
- (3)  $!F1 = !A \# !B;$

In the example above, equation (3) uses two product terms, while equation (1) requires only one. This logic function will use fewer product terms in a non-inverting device than in an inverting device. The logic function performed from input pins to output pins will be the same for both polarities.

Not all logic functions are best optimized to positive polarity. For example, the inverted form of F2, equation (3), uses fewer product terms than equation (2).

- (1)  $F2 = (A \# B) \& (C \# D);$
- (2)  $F2 = (A \& C) \# (A \& D) \# (B \& C) \# (B \& D);$
- (3)  $!F2 = (!A \& !B) \# (!C \& !D);$

Programmable polarity devices are popular because they can provide a mix of non-inverting and inverting outputs to achieve the best fit.

### Polarity Control with Istype

In ABEL-HDL, you control the polarity of the design equations and target device (in the case of programmable polarity devices) in two ways:

- Using Istype 'neg', 'pos' and 'dc'
- Using Istype 'invert' and 'buffer'

#### Using Istype 'neg', 'pos', and 'dc' to Control Equation and Device Polarity

The 'neg', 'pos', and 'dc' attributes specify types of optimization for the polarity as follows:

'neg'	<b>Istype 'neg'</b> optimizes the circuit for negative polarity. Unspecified logic in truth tables and state diagrams becomes a 0.
'pos'	<b>Istype 'pos'</b> optimizes the circuit for positive polarity. Unspecified logic in truth tables and state diagrams becomes a 1.
'dc'	<b>Istype 'dc'</b> uses polarity for best optimization. Unspecified logic in truth tables and state diagrams becomes don't care (X).

### Using 'invert' and 'buffer' to Control Programmable Inversion

An optional method for specifying the desired state of a programmable polarity output is to use the 'invert' or 'buffer' attributes. These attributes ensure that an inverter gate either does or does not exist between the output of a flip-flop and its corresponding output pin. When you use the 'invert' and 'buffer' attributes, you can still use automatic polarity selection if the target architecture features programmable inverters located before the associated flip-flop.



#### **NOTE**

The 'invert' and 'buffer' attributes do not actually control device or equation polarity — they only enforce the existence or nonexistence of an inverter between a flip-flop and its output pin.

The polarity of devices that feature a fixed inverter in this location, and a programmable inverter before the register, cannot be specified using 'invert' and 'buffer'.

## Flip-flop Equations

Pin-to-pin equations (using the := assignment operator) are only supported for D flip-flops. ABEL-HDL does not support the := assignment operator for T, SR or JK flip-flops and has no provision for specifying a particular output pin value for these types.

If you write an equation of the form:

```
Q1 := 1;
```

and the output, Q1, has been declared as a T-type flip-flop, the ABEL-HDL compiler will give a warning and convert the equation to

```
Q1.T = 1;
```

Since the T input to a T-type flip-flop does not directly correspond to the value you observed on the associated output pin, this equation will not result in the pin-to-pin behavior you want.

To produce specific pin-to-pin behavior for alternate flip-flop types, you must consider the behavior of the flip-flop you used and write detailed equations that stimulate the inputs of that flip-flop. A detailed equation to set and hold a T-type flip-flop is shown below:

$$Q1.T = !Q1.Q;$$

## Feedback Considerations — Dot Extensions

The source of feedback is normally set by the architecture of the target device. If you don't specify a particular feedback path, the design may operate differently in different device types. Specifying feedback paths (with the .FB, .Q or .PIN dot extensions) eliminates architectural ambiguities. Specifying feedback paths also allows you to use architecture-independent simulation.

The following rules should be kept in mind when you are using feedback:

- **No Dot Extension** — A feedback signal with no dot extension (for example, `count := count+1;`) results in pin feedback if it exists in the target device. If there is no pin feedback, register feedback is used, with the value of the register contents complemented (normalized) if needed to match the value observed on the pin.
- **.FB Extension** — A signal specified with the .FB extension (for example, `count := count.fb+1;`) results in register feedback normalized to the pin value if a register feedback path exists. If no register feedback is available, pin feedback is used, and the fuse mapper checks that the output enable does not conflict with the pin feedback path. If there is a conflict, an error is generated if the output enable is not constantly enabled.
- **.COM Extension** — A signal specified with the .COM extension (for example, `count := count.com+1;`) results in OR-array (pre-register) feedback, normalized to the pin value if an OR-array feedback path exists. If no OR-array feedback is available, pin feedback is used and the fuse mapper checks that the output enable does not conflict with the pin feedback path. If there is a conflict, an error is generated if the output enable is not constantly enabled.
- **.PIN Extension** — If a signal is specified with the .PIN extension (for example, `count := count.pin+1;`), the pin feedback path will be used. If the specified device does not feature pin feedback, an error will be generated. Output enables frequently affect the operation of fed-back signals that originate at a pin.
- **.Q Extension** — Signals specified with the .Q extension (for example, `count.d = count.q+1;`) will originate at the Q output of the associated flip-flop. The fed-back value may or may not correspond to the value you observe on the associated output pin; if an inverter is located between the Q output of the flip-flop and the output pin (as is the case in most registered PAL-type devices), the value of the fed-back signal will be the complement of the value you observe on the pin.

- **.D Extension** — Some devices allow feedback of the input to the register. To select this feedback, use the .D extension. Some device kits also support .COM for this feedback; refer to your device kit manual for detailed information.

## Dot Extensions and Architecture-Independence

To be architecture-independent, you must write your design in terms of its pin-to-pin behavior rather than in terms of specific device features (such as flip-flop configurations or output inversions).

For example, consider the simple circuit shown in the following (Figure 4-3). This circuit toggles high when the Toggle input is forced high, and low when the Toggle is low. The circuit also contains a three-state output enable that is controlled by the active-low Enable input.

*Dot Extensions and Architecture-independence: Circuit 1*




Figure 4-3. Dot Extensions and Architecture-independence: Circuit 1

The following simple ABEL-HDL design describes this simple one-bit synchronous circuit. The design description uses architecture-independent dot extensions to describe the circuit in terms of its behavior, as observed on the output pin of the target device. Since this design is architecture-independent, it will operate the same (disregarding initial powerup state), irrespective of the device type.

```

module pin2pin;
    Clk      pin 1;
    Toggle   pin 2;
    Ena      pin 11;
    Qout     pin 19 istype 'reg';

    equations
        Qout      := !Qout.FB & Toggle;
        Qout.CLK  = Clk;
        Qout.OE   = !Ena;

    test_vectors([Clk,Ena,Toggle] -> [Qout])
        [.c., 0 , 0 ] -> 0;
        [.c., 0 , 1 ] -> 1;
        [.c., 0 , 1 ] -> 0;
        [.c., 0 , 1 ] -> 1;
        [.c., 0 , 1 ] -> 0;
        [.c., 1 , 1 ] -> .Z.;
        [ 0 , 0 , 1 ] -> 1;
        [.c., 1 , 1 ] -> .Z.;
        [ 0 , 0 , 1 ] -> 0;
end

```

Figure 4-4. Pin-to-pin One-bit Synchronous Circuit module pin2pin

If you implement this circuit in a simple GAL16LV8 device (either by adding a device declaration statement or by specifying the P16R8 in the Fuseasm process), the result will be a circuit like the one illustrated in the following figure (Figure 4-5). Since the GAL16LV8 features inverted outputs, the design equation is automatically modified to take the feedback from Q-bar instead of Q.

*Dot Extensions and Architecture-independence: Circuit 2*




Figure 4-5. Dot Extensions and Architecture-independence: Circuit 2



## Dot Extensions and Detail Design Descriptions

You may need to be more specific about how you implement a circuit in a target device. More-complex device architectures have many configurable features, and you may want to use these features in a particular way. You may want a precise powerup and preset operation or, in some cases, you may need to control internal elements.

The circuit previously described (using architecture-independent dot extensions) could be described, for example, using detailed dot extensions in the following ABEL-HDL source file.

```

module detail1
  dl      device  'P16R8';
  Clk     pin 1;
  Toggle  pin 2;
  Ena     pin 11;
  Qout    pin 19 istype 'reg_D';

  equations
    !Qout.D    = Qout.Q & Toggle;
    Qout.CLK   = Clk;
    Qout.OE    = !Ena;

  test_vectors([Clk,Ena,Toggle] -> [Qout])
    [.c., 0 , 0 ] -> 0;
    [.c., 0 , 1 ] -> 1;
    [.c., 0 , 1 ] -> 0;
    [.c., 0 , 1 ] -> 1;
    [.c., 0 , 1 ] -> 0;
    [.c., 1 , 1 ] -> .Z.;
    [ 0 , 0 , 1 ] -> 1;
    [.c., 1 , 1 ] -> .Z.;
    [ 0 , 0 , 1 ] -> 0;

end

```

Figure 4-6. Detailed One-bit Synchronous Circuit with Inverted Qout

This version of the design will result in exactly the same fuse pattern as indicated in Figure 4-5. As written, this design assumes the existence of an inverted output for the signal Qout. This is why the Qout.D and Qout.Q signals are reversed from the architecture-independent version of the design presented earlier.



### NOTE

The inversion operator applied to Qout.D does not correspond directly to the inversion found on each output of a P16R8. The equation for Qout.D actually refers to the D input of one of the GAL16LV8's flip-flops; the output inversion found in a P16R8 is located after the register and is assumed rather than specified.

To implement this design in a device that does not feature inverted outputs, the design description must be modified. The following example shows how to write this detailed design:

```

module detail2
    Clk      pin 1;
    Toggle   pin 2;
    Ena      pin 11;
    Qout     pin 19 istype 'reg_D';

    equations
        Qout.D      = !Qout.Q & Toggle;
        Qout.CLK    = Clk;
        Qout.OE     = !Ena;

    test_vectors([Clk,Ena,Toggle] -> [Qout])
        [.c., 0 , 0 ] -> 0;
        [.c., 0 , 1 ] -> 1;
        [.c., 0 , 1 ] -> 0;
        [.c., 0 , 1 ] -> 1;
        [.c., 0 , 1 ] -> 0;
        [.c., 1 , 1 ] -> .Z.;
        [ 0 , 0 , 1 ] -> 1;
        [.c., 1 , 1 ] -> .Z.;
        [ 0 , 0 , 1 ] -> 0;
end

```

Figure 4-7. Detail One-bit Synchronous Circuit with non-inverted Qout

## Using Don't Care Optimization

Use Don't Care optimization to reduce the amount of logic required for an incompletely specified function. The @DCSET directive (used for logic description sections) and ISTYPE attribute 'dc' (used for signals) specify don't care values for unspecified logic.

Consider the following ABEL-HDL truth table:

```
truth_table ([i3,i2,i1,i0]->[f3,f2,f1,f0])
  [ 0, 0, 0, 0]->[ 0, 0, 0, 1];
  [ 0, 0, 0, 1]->[ 0, 0, 1, 1];
  [ 0, 0, 1, 1]->[ 0, 1, 1, 1];
  [ 0, 1, 1, 1]->[ 1, 1, 1, 1];
  [ 1, 1, 1, 1]->[ 1, 1, 1, 0];
  [ 1, 1, 1, 0]->[ 1, 1, 0, 0];
  [ 1, 1, 0, 0]->[ 1, 0, 0, 0];
  [ 1, 0, 0, 0]->[ 0, 0, 0, 0];
```

This truth table has four inputs, and therefore sixteen (24) possible input combinations. The function specified, however, only indicates eight significant input combinations. For each of the design outputs (f3 through f0) the truth table specifies whether the resulting value should be 1 or 0. For each output, then, each of the eight individual truth table entries can be either a member of a set of true functions called the on-set, or a set of false functions called the off-set.

Using output f3, for example, the eight input conditions can be listed as on-sets and off-sets as follows (maintaining the ordering of inputs as specified in the truth table above):

on-set of f3	off-set of f3
0 1 1 1	0 0 0 0
1 1 1 1	0 0 0 1
1 1 1 0	0 0 1 1
1 1 0 0	1 0 0 0

The remaining eight input conditions that do not appear in either the on-set or off-set are said to be members of the dc-set, as follows for f3:

```
dc-set of f3
  0 0 1 0
  0 1 0 0
  0 1 0 1
  0 1 1 0
  1 0 0 1
  1 0 1 0
  1 0 1 1
  1 1 0 1
```

Expressed as a Karnaugh map, the on-set, off-set and dc-set would appear as follows (with ones indicating the on-set, zeroes indicating the off-set, and dashes indicating the dc-set):

If the don't-care entries in the Karnaugh map are used for optimization, the function for f3 can be reduced to a single product term (f3 = i2) instead of the two (f3 = i3 & i2 & !i0 # i2 & i1 & i0) otherwise required.

The ABEL-HDL compiler uses this level of optimization if the @DCSET directive or ISTYPE 'dc' is included in the ABEL-HDL source file, as shown below.

```

module dc
  i3,i2,i1,i0      pin;
  f3,f2,f1,f0      pin istype 'dc,com';

  truth_table ([i3,i2,i1,i0]->[f3,f2,f1,f0])
    [ 0, 0, 0, 0]->[ 0, 0, 0, 1];
    [ 0, 0, 0, 1]->[ 0, 0, 1, 1];
    [ 0, 0, 1, 1]->[ 0, 1, 1, 1];
    [ 0, 1, 1, 1]->[ 1, 1, 1, 1];
    [ 1, 1, 1, 1]->[ 1, 1, 1, 0];
    [ 1, 1, 1, 0]->[ 1, 1, 0, 0];
    [ 1, 1, 0, 0]->[ 1, 0, 0, 0];
    [ 1, 0, 0, 0]->[ 0, 0, 0, 0];

end

```

Figure 4-8. Source File Showing Don't Care Optimization

This example results in a total of four single-literal product terms, one for each output. The same example (with no istype 'dc') results in a total of twelve product terms.

For truth tables, Don't Care optimization is almost always the best method. For state machines, however, you may not want undefined transition conditions to result in unknown states, or you may want to use a default state (determined by the type of flip-flops used for the state register) for state diagram simplification.

When using don't care optimization, be careful not to specify overlapping conditions (specifying both the on-set and dc-set for the same conditions) in your truth tables and state diagrams. Overlapping conditions result in an error message.

For state diagrams, you can perform additional optimization for design outputs if you specify the @dcstate attribute. If you enter @dcstate in the source file, all state diagram transition conditions are collected during state diagram processing. These transitions are then complemented and applied to the design outputs as don't-cares. You must use @dcstate in combination with @dcset or the 'dc' attribute.

## Exclusive OR Equations

Designs written for exclusive-OR (XOR) devices should contain the 'xor' attribute for architecture-independence.

### Optimizing XOR Devices

You can use XOR gates directly by writing equations that include XOR operators, or you can use implied XOR gates. XOR gates can minimize the total number of product terms required for an output or they can emulate alternate flip-flop types.

### Using XOR Operators in Equations

If you want to write design equations that include XOR operators, you must either specify a device that features XOR gates in your ABEL-HDL source file, or specify the 'xor' attribute for all output signals that will be implemented with XOR gates. This preserves one top-level XOR operator for each design output. For example,

```
module X1
    Q1      pin      istype 'com,xor';
    a,b,c   pin;
equations
    Q1 = a $ b & c;
end
```

Also, when writing equations for XOR PALs, you should use parentheses to group those parts of the equation that go on either side of the XOR. This is because the XOR operator (\$) and the OR operator (#) have the same priority in ABEL-HDL. See example **octalf.abl**.

### Using Implied XORs in Equations

High-level operators in equations often result in the generation of XOR operators. If you specify the 'XOR' attribute, these implied XORs are preserved, decreasing the number of product terms required. For example,

```
module X2
    q3,q2,q1,q0      pin istype 'reg,xor';
    clock            pin;
    count = [q3..q0];
equations
    count.clk = clock;
    count := count.FB + 1;
end
```

This design describes a simple four-bit counter. Since the addition operator results in XOR operators for the four outputs, the 'xor' attribute can reduce the amount of circuitry generated.

**NOTE**

The high-level operator that generates the XOR operators must be the top-level (lowest priority) operation in the equation. An equation such as `count := (count.FB + 1) & !reset;` does not result in the preservation of top-level XOR operators, since the `&` operator is the top-level operator.

## Using XORs for Flip-flop Emulation


Another way to use XOR gates is for flip-flop emulation. If you are using an XOR device that has outputs featuring an XOR gate and D-type flip-flops, you can write your design as if you were going to be implementing it in a device with T-type flip-flops. The XOR gates and D-type flip-flops emulate the specified T-type flip-flops. When using XORs in this way, you should not use the 'xor' attribute for output signals unless the target device has XOR gates.

### JK Flip-Flop Emulation

You can emulate JK flip-flops using a variety of circuitry found in programmable devices. When a T-type flip-flop is available, you can emulate JK flip-flops by ANDing the Q output of the flip-flop with the K input. The !Q output is then ANDed with the J input. The outputs of these two AND gates are ORed together to produce the T input of the flip-flop.

Figure 4-9 illustrates the circuitry and the Boolean expression.

*JK Flip-flop Emulation Using T Flip-flop*



$$Q := (J \& !Q) \# (K \& Q)$$

0777-1

Figure 4-9. JK Flip-flop Emulation Using T Flip-flop

You can emulate a JK flip-flop with a D flip-flop and an XOR gate. This technique is useful in devices such as the GAL20VP8. The circuitry and Boolean expression is shown in Figure 4-10.

*T Flip-flop Emulation Using D Flip-flop*




Figure 4-10. T Flip-flop Emulation Using D Flip-flop

Finally, you can also emulate a JK flip-flop by combining the D flip-flop emulation of a T flip-flop, Figure 4-10, with the circuitry of Figure 4-1. The following figure illustrates this concept.

*JK Flip-flop Emulation, D Flip-flop with XOR*




Figure 4-11. JK Flip-flop Emulation, D Flip-flop with XOR

Integrated Circuit in Digital Electronics  
Aripad Bama and Dan Porat  
John Wiley & Sons, 1973  
0755-1

## State Machines

A state machine is a digital device that traverses a predetermined sequence of states. State-machines are typically used for sequential control logic. In each state, the circuit stores its past history and uses that history to determine what to do next.

This section provides some guidelines to help you make state diagrams easy to read and maintain and to help you avoid problems. State machines often have many different states and complex state transitions that contribute to the most common problem, which is too many product terms being created for the chosen device. The topics discussed in the following subsections help you avoid this problem by reducing the number of required product terms.

The following subsections provide state machine considerations:

- Use Identifiers Rather Than Numbers for States
- Powerup Register States
- Unsatisfied Transition Conditions, D-Type Flip-Flops
- Unsatisfied Transition Conditions, Other Flip-Flops
- Number Adjacent States for a One-bit Change
- Use State Register Outputs to Identify States
- Use Symbolic State Descriptions

### Use Identifiers Rather Than Numbers for States

A state machine has different “states” that describe the outputs and transitions of the machine at any given point. Typically, each state is given a name, and the state machine is described in terms of transitions from one state to another. In a real device, such a state machine is implemented with registers that contain enough bits to assign a unique number to each state. The states are actually bit values in the register, and these bit values are used along with other signals to determine state transitions.

As you develop a state diagram, you need to label the various states and state transitions. If you label the states with identifiers that have been assigned constant values, rather than labeling the states directly with numbers, you can easily change the state transitions or register values associated with each state.

When you write a state diagram, you should first describe the state machine with names for the states, and then assign state register bit values to the state names.

For an example, see Figure 4-12 for a state machine named “sequence.” (This state machine is also discussed in the design examples.) Identifiers (A, B, and C) specify the states. These identifiers are assigned a constant decimal value in the declaration section that identifies the bit values in the state register for each state. A, B, and C are only identifiers: they do not indicate the bit pattern of the state machine. Their declared values define the value of the state register (sreg) for each state. The declared values are 0, 1, and 2.



```

module Sequence
title 'State machine example';

    q1,q0                pin    14,15 istype 'reg';
    clock,enab,start,hold,reset pin  1,11,4,2,3;
    halt                 pin    17 istype 'reg';
    in_B,in_C           pin    12,13 istype 'com';
    sreg                 =      [q1,q0];

"State Values...
    A = 0;              B = 1;              C = 2;

equations
    [q1,q0,halt].clk = clock;
    [q1,q0,halt].oe  = !enab;
state_diagram sreg;
    State A:           " Hold in state A until start is active.
        in_B = 0;
        in_C = 0;
        IF (start & !reset) THEN B WITH halt := 0;
        ELSE A WITH halt := halt.fb;

    State B:           " Advance to state C unless reset is active
        in_B = 1;           " or hold is active. Turn on halt indicator
        in_C = 0;           " if reset.
        IF (reset) THEN A WITH halt := 1;
        ELSE IF (hold) THEN B WITH halt := 0;
        ELSE C WITH halt := 0;

    State C:           " Go back to A unless hold is active
        in_B = 0;           " Reset overrides hold.
        in_C = 1;
        IF (hold & !reset) THEN C WITH halt := 0;
        ELSE A WITH halt := 0;

test_vectors([clock,enab,start,reset,hold]->[sreg,halt,in_B,in_C])
    [ .p. , 0 , 0 , 0 , 0 ]->[ A , 0 , 0 , 0 ];
    [ .c. , 0 , 0 , 0 , 0 ]->[ A , 0 , 0 , 0 ];
    [ .c. , 0 , 1 , 0 , 0 ]->[ B , 0 , 1 , 0 ];
    [ .c. , 0 , 0 , 0 , 0 ]->[ C , 0 , 0 , 1 ];

    [ .c. , 0 , 1 , 0 , 0 ]->[ A , 0 , 0 , 0 ];
    [ .c. , 0 , 1 , 0 , 0 ]->[ B , 0 , 1 , 0 ];
    [ .c. , 0 , 0 , 1 , 0 ]->[ A , 1 , 0 , 0 ];
    [ .c. , 0 , 0 , 0 , 0 ]->[ A , 1 , 0 , 0 ];

    [ .c. , 0 , 1 , 0 , 0 ]->[ B , 0 , 1 , 0 ];
    [ .c. , 0 , 0 , 0 , 1 ]->[ B , 0 , 1 , 0 ];
    [ .c. , 0 , 0 , 0 , 1 ]->[ B , 0 , 1 , 0 ];
    [ .c. , 0 , 0 , 0 , 0 ]->[ C , 0 , 0 , 1 ];

end

```

Figure 4-12. Using Identifiers for States

## Powerup Register States

If a state machine has to have a specific starting state, you must define the register powerup state in the state diagram description or make sure your design goes to a known state at powerup. Otherwise, the next state is undefined.

## Unsatisfied Transition Conditions

### D-Type Flip-Flops

For each state described in a state diagram, you specify the transitions to the next state and the conditions that determine those transitions. For devices with D-type flip-flops, if none of the stated conditions are met, the state register, shown in the following figure, is cleared to all 0s on the next clock pulse. This action causes the state machine to go to the state that corresponds to the cleared state register. This can either cause problems or you can use it to your advantage, depending on your design.

*D-type Register with False Inputs*




Figure 4-13. D-type Register with False Inputs

You can use the clearing behavior of D-type flip-flops to eliminate some conditions in your state diagram, and some product terms in the converted design, by leaving the cleared-register state transition implicit. If no specified transition condition is met, the machine goes to the cleared-register state. This behavior can also cause problems if the cleared-register state is undefined in the state diagram, because if the transition conditions are not met for any state, the machine goes to an undefined state and stays there.

To avoid problems caused by this clearing behavior, always have a state assigned to the cleared-register state. Or, if you do not assign a state to the cleared-register state, define every possible condition so some condition is always met for each state. You can also use the automatic transition to the cleared-register state by eliminating product terms and explicit definitions of transitions. You can also use the cleared-register state to satisfy illegal conditions.

## Other Flip-flops

If none of the state conditions is met in a state machine that employs JK, RS, and T-type flip-flops, the state machine does not advance to the next state, but holds its present state due to the low input to the register from the OR array output. In such a case, the state machine can get stuck in a state. You can use this holding behavior to your advantage in some designs.

## Precautions for Using Don't Care Optimization

When you use don't care optimization, you need to avoid certain design practices. The most common design technique that conflicts with this optimization is mixing equations and state diagrams to describe default transitions. For example, consider the design shown in the following figure.

```

module TRAFFIC
title 'Traffic Signal Controller'

    Clk,SenA,SenB    pin  1, 8, 7;
    PR               pin  16;           "Preset control
    GA,YA,RA        pin  15..13;
    GB,YB,RB        pin  11..9;

    "Node numbers are not required if fitter is used
    S3..S0          node 31..34 istype 'reg_sr,buffer';
    COMP            node 43;

    H,L,Ck,X        = 1, 0, .C., .X.;
    Count           = [S3..S0];

    "Define Set and Reset inputs to traffic light flip-flops
    GreenA = [GA.S,GA.R];
    YellowA = [YA.S,YA.R];
    RedA    = [RA.S,RA.R];
    GreenB  = [GB.S,GB.R];
    YellowB = [YB.S,YB.R];
    RedB    = [RB.S,RB.R];
    On      = [ 1 , 0 ];
    Off     = [ 0 , 1 ];

    " test_vectors edited
    equations
        [GB,YB,RB].AP = PR;
        [GA,YA,RA].AP = PR;
        [GB,YB,RB].CLK = Clk;
        [GA,YA,RA].CLK = Clk;
        [S3..S0].AP = PR;
        [S3..S0].CLK = Clk;

```

Figure 4-14. State Machine Description with Conflicting Logic

```

"Use Complement Array to initialize or restart
[S3..S0].R      = (!COMP & [1,1,1,1]);
[GreenA,YellowA,RedA] = (!COMP & [On ,Off,Off]);
[GreenB,YellowB,RedB] = (!COMP & [Off,Off,On ]);

state_diagram Count
State 0:      if ( SenA & !SenB ) then 0 with COMP = 1;
              if (!SenA & SenB ) then 4 with COMP = 1;
              if ( SenA == SenB ) then 1 with COMP = 1;

State 1:      goto  2 with COMP = 1;
State 2:      goto  3 with COMP = 1;
State 3:      goto  4 with COMP = 1;
State 4:      GreenA  = Off;
              YellowA = On ;
              goto  5 with COMP = 1;

State 5:      YellowA = Off;
              RedA    = On ;
              RedB    = Off;
              GreenB  = On ;
              goto  8 with COMP = 1;

State 8:      if (!SenA & SenB ) then  8 with COMP = 1;
              if ( SenA & !SenB ) then 12 with COMP = 1;
              if ( SenA == SenB ) then  9 with COMP = 1;

State 9:      goto 10 with COMP = 1;
State 10:     goto 11 with COMP = 1;
State 11:     goto 12 with COMP = 1;
State 12:     GreenB  = Off;
              YellowB = On ;
              goto 13 with COMP = 1;

State 13:     YellowB = Off;
              RedB    = On ;
              RedA    = Off;
              GreenA  = On ;
              goto  0 with COMP = 1;

end

```

Figure 4-14 State Machine Description with Conflicting Logic (Continued)

This design uses the complement array feature of the Signetics FPLA devices to perform an unconditional jump to state [0,0,0,0]. If you use the **@DCSET** directive, the equation that specifies this transition

```
[S3,S2,S1,S0].R = (!COMP & [1,1,1,1]);
```

will conflict with the dc-set generated by the state diagram for S3.R, S2.R, S1.R, and S0.R. If equations are defined for state bits, the **@DCSET** directive is incompatible. This conflict would result in an error and failure when the logic for this design is optimized.

To correct the problem, you must remove the **@DCSET** directive so the implied dc-set equations are folded into the off-set for the resulting logic function. Another option is to rewrite the module as shown below.

```
module TRAFFIC1
title 'Traffic Signal Controller'

      Clk,SenA,SenB   pin   1, 8, 7;
      PR              pin   16;           "Preset control
      GA,YA,RA       pin   15..13;
      GB,YB,RB       pin   11..9;

      S3..S0         node 31..34 istype 'reg_sr,buffer';
      H,L,Ck,X       = 1, 0, .C., .X.;
      Count          = [S3..S0];

"Define Set and Reset inputs to traffic light flip flops
GreenA = [GA.S,GA.R];
YellowA = [YA.S,YA.R];
RedA    = [RA.S,RA.R];
GreenB  = [GB.S,GB.R];
YellowB = [YB.S,YB.R];
RedB    = [RB.S,RB.R];
On      = [ 1 , 0 ];
Off     = [ 0 , 1 ];

" test_vectors edited
equations
  [GB,YB,RB].AP = PR;
  [GA,YA,RA].AP = PR;
  [GB,YB,RB].CLK = Clk;
  [GA,YA,RA].CLK = Clk;
  [S3..S0].AP = PR;
  [S3..S0].CLK = Clk;
```

Figure 4-15. @DCSET-compatible State Machine Description

```

@DCSET
state_diagram Count
    State 0:      if ( SenA & !SenB ) then 0;
                  if (!SenA & SenB ) then 4;
                  if ( SenA == SenB ) then 1;

    State 1:      goto 2;
    State 2:      goto 3;
    State 3:      goto 4;

    State 4:      GreenA = Off;
                  YellowA = On ;
                  goto 5;

    State 5:      YellowA = Off;
                  RedA    = On ;
                  RedB    = Off;
                  GreenB  = On ;
                  goto 8;

    State 6:      goto 0;
    State 7:      goto 0;

    State 8:      if (!SenA & SenB ) then 8;
                  if ( SenA & !SenB ) then 12;
                  if ( SenA == SenB ) then 9;

    State 9:      goto 10;
    State 10:     goto 11;
    State 11:     goto 12;
    State 12:     GreenB = Off;
                  YellowB = On ;
                  goto 13;

    State 13:     YellowB = Off;
                  RedB    = On ;
                  RedA    = Off;
                  GreenA  = On ;
                  goto 0;

    State 14:     goto 0;
    State 15:     "Power up and preset state
                  RedA    = Off;
                  YellowA = Off;
                  GreenA  = On ;
                  RedB    = On ;
                  YellowB = Off;
                  GreenB  = Off;
                  goto 0;

end

```

Figure 4-15 @DCSET-compatible State Machine Description (Continued)

## Number Adjacent States for One-bit Change

You can reduce the number of product terms produced by a state diagram by carefully choosing state register bit values. Your state machine should be described with symbolic names for the states, as described above. Then, if you assign the numeric constants to these names so the state register bits change by only one bit at a time as the state machine goes from state to state, you will reduce the number of product terms required to describe the state transitions.

As an example, take the states A, B, C, and D, which go from one state to the other in alphabetical order. The simplest choice of bit values for the state register is a numeric sequence, but this is not the most efficient method. To see why, examine the following bit value assignments. The preferred bit values cause a one-bit change as the machine moves from state B to C, whereas the simple bit values cause a change in both bit values for the same transition. The preferred bit values produce fewer product terms.

State	Simple Bit Values	Preferred Bit Values
A	00	00
B	01	01
C	10	11
D	11	10

If one of your state register bits uses too many product terms, try reorganizing the bit values so that state register bit changes in value as few times as possible as the state machine moves from state to state.

Obviously, the choice of optimum bit values for specific states can require some tradeoffs; you may have to optimize for one bit and, in the process, increase the value changes for another. The object should be to eliminate as many product terms as necessary to fit the design into the device.

## Use State Register Outputs to Identify States

Sometimes it is necessary to identify specific states of a state machine and signal an output that the machine is in one of these states. Fewer equations and outputs are needed if you organize the state register bit values so one bit in the state register determines if the machine is in a state of interest. Take, for example, the following sequence of states in which identification of the  $Cn$  states is required:

**State Register Bit Values**

<b>State Name</b>	<b>Q3</b>	<b>Q2</b>	<b>Q1</b>
A	0	0	0
B	0	0	1
C1	1	0	1
C2	1	1	1
C3	1	1	0
D	0	1	0

This choice of state register bit values allows you to use Q3 as a flag to indicate when the machine is in any of the  $C_n$  states. When Q3 is high, the machine is in one of the  $C_n$  states. Q3 can be assigned directly to an output pin on the device. Notice also that these bit values change by only one bit as the machine cycles through the states, as is recommended in the section above.



## Using Symbolic State Descriptions

Symbolic state descriptions describe a state machine without having to specify actual state values. A symbolic state description is shown below.

```

module SM
  a,b,clock      pin;           " inputs
  a_reset,s_reset pin;         " reset inputs
  x,y            pin istance 'com'; " simple outputs

  sreg1          state_register;
  S0..S3        state;

equations
  sreg1.clk = clock;

state_diagram sreg1
  state S0:
    goto S1 with {x = a & b;
                  y = 0;   }
  state S1: if (a & b)
            then S2 with {x = 0;
                          y = 1; }
  state S2: x = a & b;
            y = 1;
            if (a) then S1 else S2;
  state S3:
    goto S0 with {x = 1;
                  y = 0; }

  async_reset S0: a_reset;
  sync_reset  S0: s_reset;
end

```

Figure 4-16. Symbolic State Description

Symbolic state descriptions use the same syntax as non-symbolic state descriptions; the only difference is the addition of the **STATE\_REGISTER** and **STATE** declarations, and the addition of symbolic synchronous and asynchronous reset statements.

### Symbolic Reset Statements

In symbolic state descriptions, the **SYNC\_RESET** and **ASYNC\_RESET** statements specify synchronous or asynchronous state machine reset logic. For example, to specify that a state machine must asynchronously reset to state Start when the Reset input is true, you write

```
ASYNC_RESET Start : (Reset) ;
```

## Symbolic Test Vectors

You can also write test vectors to refer to symbolic state values by entering the symbolic state register name in the test vector header (in the output sections), and the symbolic state names in the test vectors as output values.

## Using Complement Arrays

The complement array is a unique feature found in some logic sequencers. This section shows a typical use ending counter sequence.

You can use transition equations to express the design of counters and state machines in some devices with JK or SR flip-flops. A transition equation expresses a state of the circuit as a variation of, or adjustment to, the previous state. This type of equation eliminates the need to specify every node of the circuit; you can specify only those that require a transition to the opposite state.

An example of transition equations is shown in Figure 4-17, a source file for a decade counter having a single (clock) input and a single latched output. This counter divides the clock input by a factor of ten and generates a 50% duty-cycle squarewave output. In addition to its registered outputs, this device contains a set of “buried” (or feedback) registers whose outputs are fed back to the product term inputs. These nodes must be declared, and can be given any names.

Node 49, the complement array feedback, is declared (as COMP) so that it can be entered into each of the equations. In this design, the complement array feedback is used to wrap the counter back around to zero from state nine, and also to reset it to zero if an illegal counter state is encountered. Any illegal state (and also state 9) will result in the absence of an active product term to hold node 49 at a logic low. When node 49 is low, product term 9 resets each of the feedback registers so the counter is set to state zero. (To simplify the following description of the equations in Figure 4-17, node 49 and the complement array feedback are temporarily ignored.)

The first equation states that the F0 (output) register is set (to provide the counter output) and the P0 register is set when registers P0, P1, P2, and P3 are all reset (counter at state zero) and the clear input is low. The complemented outputs of the registers (with the clear input low) form product term 0. Product term 0 sets register P0 to increment the decade counter to state 1, and sets register F0 to provide an output at pin 18.

```

module DECADE
title 'Decade Counter    Uses Complement Array
Michael Holley    Data I/O Corp'

    decade            device 'F105';

    Clk,Clr,F0,PR    pin  1,8,18,19;
    P3..P0          node 40..37;
    COMP            node 49;

    F0,P3..P0    istype 'reg_sr,buffer';
    _State       = [P3,P2,P1,P0];
    H,L,Ck,X     = 1, 0, .C., .X.;

equations
    [P3,P2,P1,P0,F0].ap = PR;
    [F0,P3,P2,P1,P0].clk = Clk;

"Output          Next State          Present State          Input
[F0.S, COMP,    P0.S] = !P3.Q & !P2.Q & !P1.Q & !P0.Q & !Clr; "0 to 1
[    COMP,      P1.S,P0.R] = !P3.Q & !P2.Q & !P1.Q & P0.Q & !Clr; "1 to 2
[    COMP,      P0.S] = !P3.Q & !P2.Q & P1.Q & !P0.Q & !Clr; "2 to 3
[    COMP,    P2.S,P1.R,P0.R] = !P3.Q & !P2.Q & P1.Q & P0.Q & !Clr; "3 to 4
[    COMP,      P0.S] = !P3.Q & P2.Q & !P1.Q & !P0.Q & !Clr; "4 to 5
[F0.R, COMP,    P1.S,P0.R] = !P3.Q & P2.Q & !P1.Q & P0.Q & !Clr; "5 to 6
[    COMP,      P0.S] = !P3.Q & P2.Q & P1.Q & !P0.Q & !Clr; "6 to 7
[    COMP,P3.S,P2.R,P1.R,P0.R] = !P3.Q & P2.Q & P1.Q & P0.Q & !Clr; "7 to 8
[    COMP      P0.S] = P3.Q & !P2.Q & !P1.Q & !P0.Q & !Clr; "8 to 9
[      P3.R,P2.R,P1.R,P0.R] =                                !COMP; "Clear

"After Preset, clocking is inhibited until High-to-Low clock transition.
test_vectors ([Clk,PR,Clr] -> [_State,F0 ])
    [ 0 , 0, 0 ] -> [  X , X];
    [ 1 , 1, 0 ] -> [^b1111, H]; " Preset high
    [ 1 , 0, 0 ] -> [^b1111, H]; " Preset low
    [ Ck, 0, 0 ] -> [  0 , H]; " COMP forces to State 0
    [ Ck, 0, 0 ] -> [  1 , H];
"
    ..vectors edited...
    [ Ck, 0, 1 ] -> [  0 , H]; " Clear
end

```

Figure 4-17. Transition Equations for a Decade Counter

The second equation performs a transition from state 1 to state 2 by setting the P1 register and resetting the P0 register. (The .R dot extension is used to define the reset input of the registers.) In state 2, the F0 register remains set, maintaining the high output. The third equation again sets the P0 register to achieve state 3 (P0 and P1 both set), while the fourth equation resets P0 and P1, and sets P2 for state 4, and so on.

Wraparound of the counter from state 9 to state 0 is achieved by means of the complement array node (node 49). The last equation defines state 0 (P3, P2, P1, and P0 all reset) as equal to !COMP, that is, node 49 at a logic low. When this equation is processed, the fuses are blown. As a result, the !COMP signal is true to generate product term 9 and reset all the “buried” registers to zero.

## ABEL-HDL and Truth Tables

Truth Tables in ABEL-HDL represent a very easy and straightforward description method, well suited in a number of situations involving combinational logic.

The principle of the Truth Table is to build an exhaustive list of the input combinations (referred to as the ON-set) for which the output(s) become(s) active.

The following list summarizes design considerations for Truth Tables. Following the list are more detailed examples.

- The OFF-set lines in a Truth Table are necessary when more than one output is assigned in the Truth Table. In this case, not all Outputs are fired under the same conditions, and therefore OFF-set conditions do exist.
- OFF-set lines are ignored because they represent the default situation, unless the output variable is declared dc. In this case, a third set is built, the DC-set and the Output inside it is assigned proper values to achieve the best logic reduction possible.
- If output type dc (or @dcset) is not used and multiple outputs are specified in a Truth table, consider the outputs one by one and ignore the lines where the selected output is not set.
- Don't Cares (.X.) used on the right side of a Truth Table have no optimization effect.
- When dealing with multiple outputs of different kind, avoid general settings like @DCSET which will affect all your outputs. Use istype '.....DC' on outputs for which this reduction may apply.
- Beware of Outputs for which the ON-set might be empty.
- As a general guideline, it is important not to rely on first impression or simple intuition to understand Truth tables. The way they are understood by the compiler is the only possible interpretation. This means that Truth Tables should be presented in a clear and understandable format, should avoid side effects, and should be properly documented (commented).

## Basic Syntax - Simple Examples

In this example, the lines commented as L1 and L2 are the ON-set.

Lines L3 and L4 are ignored because Out is type *default* (meaning '0' for unspecified combinations). The resulting equation does confirm this.

```

MODULE DEMO1
TITLE 'Example 1'
" Inputs
  A, B, C  pin;
"Output
  Out  pin  istype 'com';
Truth_Table
  ([A,B,C] -> Out )
  [0,1,0] -> 1;  // L1
  [1,1,1] -> 1;  // L2
  [0,0,1] -> 0;  // L3
  [1,0,0] -> 0;  // L4
END
// Resulting Reduced Equation :
// Out = (!A & B & !C) # (A & B & C);

```

Example 2 differs from example 1 because Out is now type 'COM, DC'. (optimizable don't care).

In this case, the lines commented as L1 and L2 are the ON-set, L3 and L4 are the *OFF-set* and other combinations become *don't care* (DC-set) meaning 0 or 1 to produce the best logic reduction. As a result in this example, the equation is VERY simple.

@DCSET instruction would have produced the same result as to declare Out of type dc. But @DCSET must be used with care when multiple outputs are defined: they all become dc.

```

MODULE DEMO1
TITLE 'Example 2'
" Inputs
  A, B, C  pin;
"Output
  Out  pin  istype 'com, dc';
Truth_Table
  ([A,B,C] -> Out )
  [0,1,0] -> 1;  // L1
  [1,1,1] -> 1;  // L2
  [0,0,1] -> 0;  // L3
  [1,0,0] -> 0;  // L4
END
// Resulting Reduced Equation :
// Out = (B);

```

## Influence of Signal polarity

We will see now with example 3 how the polarity of the signal may influence the truth table:

In this example, Out1 and Out2 are strictly equivalent. For !Out1, note that the ON-set is the 0 values. *The third line L3 is ignored.*

```

MODULE DEMO2
TITLE 'Example 3'
" Inputs
  A, B, C  pin;
"Output
  Out1pin istype 'com, neg';
  Out2pin istype 'com, neg';
  Out3pin istype 'com, neg'; // BEWARE

Truth_Table
  ([A,B,C] -> [!Out1, Out2, Out3] )
  [0,0,1] -> [ 0,      1,      0 ];//L1
  [0,1,1] -> [ 0,      1,      0 ];//L2
  [1,1,0] -> [ 1,      0,      1 ];//L3
END
// Resulting Equations :
//      !Out1 = !Out2 = (A # !C);
// or:  Out1 =  Out2 = (!A & C);
// BUT: Out3 = (A & B & !C); <<what you wanted ?

```

For active-low outputs, one must be careful to specify **1** for the active state if the Output appears without the exclamation point (!).

**0** must be used when !output is defined in the table header.

We recommend the style used for Out1.

For Out3, line used is L3, L1 and L2 are ignored.

## Using .X. in Truth tables conditions

Don't Care used on the left side in Truth tables have no optimization purpose. they only serve as a shortcut to write several conditions in one single line.

Be careful when using .X. in conditions. This can lead to overlapping conditions which look not consistent (see example below). Due to the way the compiler work, this type of inconsistency is not checked nor reported. In fact, only the ON-set condition is taken into account, the OFF-set condition is ignored.

The following example illustrates this:

```

MODULE DEMO3
TITLE 'Example 4'
" Inputs
  A, B, C  pin;
"Output
  Outpin istype 'com';
" Equivalence
  X = .X.;
Truth_Table
  ([A,B,C] -> Out )
  [0,0,1] ->  0; //L1 ignored in fact
  [0,1,0] ->  1; //L2
  [1,X,X] ->  1; //L3
  [0,0,1] ->  1; //L4 incompatible
  [1,1,0] ->  0; //L5 incompatible
END
// Result : Out = A # (B & !C) # (!B & C)

```

L1 is in fact ignored. Out is active high, therefore only line L4 is taken into account.

Likewise, L5 intersects L3, but is ignored since it is not in the ON-set for Out.

Globally, only L2, L3 and L4 are taken into account, as we can check in the resulting equation, without any error reported.

## Using .X. on the right side

The syntax allows to use .X. as a target value for an output. In this case, the condition is simply ignored.



### NOTE

This is **not** the method to specify optimizable don't care states. See example 2 for such an example.

Example 6 shows that-> .X. states are not optimized if DC type or @DCSET are not used.

These lines are ALWAYS ignored.

```

MODULE DEMO6
TITLE 'Example 6'
" Inputs
  A, B, C  pin;
"Output
  Outpin istype 'com';
" Equivalence
  X = .X.;
Truth_Table
  ([A,B,C] -> Out )
  [0,0,0] -> 0;
  [0,0,1] -> X;
  [0,1,0] -> 1;
  [0,1,1] -> X;
  [1,X,X] -> X;
END
// As is : Out = (!A & B & !C);
// With istype 'com,DC' : Out = (B);

```

They are in fact of no use, except maybe as a way to document that output does not matter.



## Special case: Empty ON-set

There is a special case which is unlikely to happen, but may sometimes occur. Consider this example:

```

MODULE DEMO5
TITLE 'Example 5'
" Inputs
  A, B, C pin;
"Output
  Outpin istype 'com, pos';
Truth_Table
  ([A,B,C] -> Out )
  [0,0,1] -> 0;
  [0,1,0] -> 0;
  [1,0,0] -> 0;
// [0,0,0] -> 1;//changes everything!
END
// Without the last line L4 :
// !Out=(A & !B & !C)# (!A & B & !C)# (!A & !B & C);
// WITH L4 : Out = (!A & !B & !C);

```

What we obtain is slightly unexpected. This table should produce Out=0; as the result. (We enumerated only OFF conditions, and the polarity is POS (or default), so unlisted cases should also turn into zeroes.)

One reason to build such a table could be when multiple outputs are defined, and when Out needs to be shut off for whatever reason.

In the absence of the line L4, the result is not intuitive. The output is 0 only for the listed cases (L1, L2, L3), and is 1 for all other cases, even if dc or pos is used.

When line L4 is restored, then the output equation becomes  $Out = (!A \& !B \& !C)$ ; because we fall in the general situation where the ON-set is not empty.

## Registered Logic in Truth tables

Truth Tables can specify registered outputs. In this case, the assignment becomes  $:\>$  (instead of  $\rightarrow$ ).

For more information, refer to the [ABEL-HDL Reference Manual](#).

# Index

---

## Symbols

'attribute'  
  and polarity control [54](#)

'collapse'  
  selective collapsing [42](#)

'neg'  
  and polarity control [53](#)

.D [56](#)

.FB [55](#)

.PIN [55](#)

.Q [55](#)

:=  
  alternate flip-flop types [50](#)

@DCSET  
  example [61](#)  
  with state machines [68](#)

'xor' [62](#)

“collapse”  
  collapsing nodes [42](#)

“Keep”  
  collapsing nodes [42](#)

## A

ABEL-HDL  
  enter an ABEL-HDL description [25](#)  
  enter logic description [27](#)  
  enter test vectors [28](#)  
  overview [14](#)  
  properties [31](#)  
  strategies [32](#)

ABEL-HDL Compiling [24](#)

Active-low declarations [51](#)

actlow1.abl [52](#)

actlow2.abl [51](#)

Attributes  
  and architecture independence [43](#)

Architecture independence  
  attributes [43](#)  
  dot extensions [43](#) [56](#)  
  dot extensions, example [57](#)  
  resolving ambiguities [44](#)

Arrays, complement [75](#)

Attributes  
  collapsing nodes [42](#)  
  in lower-level sources [39](#)

Auto-update [29](#)

## B

Bottom-up design [20](#)

## C

Collapsing nodes [42](#)  
  selective [42](#)

Combinational nodes [40](#)

Compilation [17](#)

Complement arrays [75](#)  
  example [76](#)

## D

D flip-flop  
  unsatisfied transition conditions [67](#)

Dangling nodes [40](#)

dc  
  and polarity control [53](#)

dc.abl [61](#)

Dc-set [60](#)  
  and optimization [61](#)

decade.abl [76](#)

Declarations  
  active-low [51](#)

Design hierarchy [17](#)

Design Overview  
  compilation [17](#)  
  device programming [17](#)  
  hierarchy [17](#)  
  projects [15](#)  
  simulation [17](#)  
  sources [16](#)

Dot extensions  
  and detail descriptions [58](#)

Detail descriptions [45](#)  
  and macrocells [45](#)  
  example, dot extensions [58](#) [59](#)  
  example, inverting [48](#)  
  example, non-inverting [47](#)  
  when to use [50](#)

detail1.abl [58](#)  
 detail2.abl [59](#)  
 Device programming [17](#)  
 Devices  
   programmable polarity [53](#)  
 Don't Care .X.  
   on left side of Truth Table [80](#)  
   on right side of Truth Table [81](#)  
 Detail descriptions  
   and dot extensions [58](#)  
 Dot extensions  
   .D [56](#)  
   .FB [55](#)  
   .PIN [55](#)  
   .Q [55](#)  
   and architecture independence [43, 56](#)  
   and architecture independence,  
     example [57](#)  
   and feedback [55](#)  
   example, detail [58, 59](#)  
   no [55](#)

**E**  
 Emulation of flip-flops [63](#)  
 Equation polarity [53](#)  
 Equations  
   for flip-flops [54](#)  
   XOR [62](#)

**F**  
 Feedback  
   and dot extensions [55](#)  
   merging [41](#)  
 Flip-flops  
   and dot extensions [54](#)  
   detail descriptions [50](#)  
   D-type [67](#)  
   emulation with XORs [63](#)  
   state diagrams [50](#)  
   using := with [50](#)

**H**  
 Hierarchical design  
   abstract [19](#)  
   advantages of [19](#)  
   approaches to [19](#)  
   bottom-up [20](#)  
   defined for ABEL-HDL [20](#)  
   mixed [20](#)  
   philosophy [19](#)  
   symbols in [20](#)

  techniques [19](#)  
   top-down [20](#)  
 Hierarchical levels  
   defined [18](#)  
 Hierarchy [17, 38](#)  
   modular design [18, 19](#)

**I**  
 Identifiers  
   in state machines [65](#)  
 Inside-out design [20](#)  
 Instantiation [38](#)  
 Interface  
   submodule [39](#)  
 Istype, and polarity control [54](#)

**J**  
 JK flip-flop  
   and := [50](#)  
   emulation of [63](#)

**L**  
 Linking modules  
   merging feedbacks [41](#)  
   post-linked optimization [41](#)  
 Lower-level sources [39](#)  
   instantiating [38](#)

**M**  
 Mixed design [20](#)

**N**  
 Node  
   collapsing [42](#)  
   combinational [40](#)  
   complement arrays [75](#)  
   dangling [40](#)  
   registered [40](#)  
   removing redundant [41](#)  
   selective collapsing [42](#)

**O**  
 Off-set [60](#)  
 One-bit changes [72](#)  
 On-set [60](#)  
   in Truth Tables [78](#)  
 Optimization  
   and @DCSET [61](#)  
   of XORs [62](#)  
   post-linked [41](#)  
   reducing product terms [72](#)  
 Output enables [39](#)

**P**

pin2pin.abl [57](#)  
 Pin-to-pin descriptions [44](#)  
   and flip-flops [54](#)  
   example [47](#)  
   resolving ambiguities [44](#)  
 Polarity control [53](#)  
   active levels [53](#)  
 Ports  
   declaring lower-level [39](#)  
 Post-linked Optimization [41](#)  
 Powerup state [67](#)  
 Preset  
   built-in, example [48](#)  
 Product terms  
   reducing [72](#)  
 Programmable designing [12](#)  
 Programmable polarity, active levels for  
   devices [53](#)  
 Project sources [16](#)  
 Properties [31](#)

**Q**

Q11.abl [47](#)  
 Q12.abl [47](#)  
 Q13.abl [48](#)  
 Q15.abl [49](#)  
 Q17.abl [49](#)

**R**

Redundant nodes [41](#)  
 Registered design descriptions [44](#)  
 Registered nodes [40](#)  
 Registers  
   bit values in state machines [72](#)  
   cleared state in state machines [67](#)  
   powerup states [67](#)  
 Reset  
   example, inverted architecture [49](#)  
   example, non-inverted architecture [49](#)  
   resolving ambiguities [49](#)

**S**

Selective collapsing [42](#)  
 sequence.abl [66](#)  
 Simulation [17](#)  
 Sources  
   ABEL-HDL [16](#)  
   device [16](#)  
   graphic waveform stimulus [16](#)  
   project notebook [16](#)  
   schematic [16](#)

test vector [16](#)  
 Verilog HDL [16](#)  
 Verilog test fixture [16](#)  
 VHDL [16](#)  
 VHDL test bench [16](#)  
 SR flip-flop  
   and := [50](#)  
 State machine example [66](#)  
   @DCSET [70](#)  
   no @DCSET [68](#)  
 State machines  
   and @DCSET [61](#), [68](#)  
   cleared register state [67](#)  
   design considerations [65](#)  
   identifiers in [65](#)  
   identifying states [72](#)  
   illegal states [67](#)  
   powerup register states [67](#)  
   reducing product terms [72](#)  
   using state register outputs [72](#)  
 State registers [72](#)  
 Strategies [32](#)  
 Symbolic state descriptions [74](#)

**T**

T flip-flop  
   and equations [54](#)  
 Top-down design [20](#)  
 traffic.abl [68](#)  
 traffic1.abl [70](#)  
 Transferring designs [43](#)  
 Transition conditions [67](#)  
 Tristate outputs [39](#)  
 Truth Tables  
   ABEL-HDL [77](#)

**X**

x1.abl [62](#)  
 x2.abl [62](#)  
 XORs  
   and operator priority [63](#)  
   example [62](#)  
   flip-flop emulation [63](#)  
   implied [62](#)  
   optimization of [62](#)